module mod_segment_tree
  implicit none
  type t_segment_tree
    private
    integer :: deflt = 1000000000
    integer :: n, p
    integer, pointer :: arr(:) => null()
  contains
    procedure :: init => init_segtree
    procedure :: release => release_segtree
    procedure :: update => update
    procedure :: query => query
    procedure :: default => set_default
    procedure :: change => change_default
  end type
  interface segment_tree
    module procedure :: newst
  end interface segment_tree
contains
  integer function op(x,y)
    integer, intent(in) :: x, y
    op = min(x,y)
  end
  function newst(n) result(ret)
    integer, intent(in) :: n
    type(t_segment_tree) :: ret
    integer :: p
    p = 1
    do while (p < n)
      p = 2*p
    end do
    ret%n = n
    ret%p = p
    allocate(ret%arr(2*p-1))
    ret%arr = ret%deflt
  end
  subroutine init_segtree(this,n)
    class(t_segment_tree), intent(inout) :: this
    integer, intent(in) :: n
    integer :: p
    p = 1
    do while (p < n)
      p = 2*p
    end do
    this%n = n
    this%p = p
    allocate(this%arr(2*p-1))
    this%arr = this%deflt
  end
  subroutine release_segtree(this)
    class(t_segment_tree), intent(inout) :: this
    if (associated(this%arr)) deallocate(this%arr)
  end
  subroutine set_default(this,i)
    class(t_segment_tree), intent(inout) :: this
    integer, intent(in) :: i
    integer :: x
    x = i+this%p-1
    this%arr(x) = this%deflt
    do while (x > 1)
      x = x/2
      this%arr(x) = op(this%arr(2*x),this%arr(2*x+1))
    end do
  end
  subroutine change_default(this,deflt)
    class(t_segment_tree), intent(inout) :: this
    integer, intent(in) :: deflt
    this%deflt = deflt
  end
  subroutine update(this,i,v)
    class(t_segment_tree), intent(inout) :: this
    integer, intent(in) :: i
    integer, intent(in) :: v
    integer :: x
    x = i+this%p-1
    this%arr(x) = v
    do while (x > 1)
      x = x/2
      this%arr(x) = op(this%arr(2*x),this%arr(2*x+1))
    end do
  end
  integer function query(this,a,b)
    class(t_segment_tree), intent(inout) :: this
    integer, intent(in) :: a, b
    integer :: l, r
    query = this%deflt
    l = a+this%p-1
    r = b+this%p-1
    do while (l <= r)
      if (mod(l,2) == 1) then
        query = op(query,this%arr(l))
        l = l+1
      end if
      if (mod(r,2) == 0) then
        query = op(query,this%arr(r))
        r = r-1
      end if
      l = l/2
      r = r/2
    end do
  end
end module mod_segment_tree
program Sugoroku
  use mod_segment_tree
  implicit none
  integer :: n, m, i, x, y, a(0:100001) = 0
  character(100001) :: s
  type(t_segment_tree) :: st
  read(*,*) n, m
  if (n <= m) then
    write(*,'(i0)') n
    stop
  end if
  read(*,*) s
  st = segment_tree(n+1)
  call st%update(1,0)
  do i = 2, n+1
    if (s(i:i) == "1") cycle
    x = st%query(max(1,i-m),i-1)
    if (x > n) then
      write(*,'(i0)') -1
      stop
    end if
    call st%update(i,x+1)
  end do
  y = st%query(n+1,n+1)
  a(y) = n+1
  do i = n+1, 1, -1
    if (s(i:i) == "1") cycle
    x = st%query(i,i)
    if (x == y) cycle
    if (i < a(x+1)-m) cycle
    a(x) = i
  end do
  write(*,'(i0)',advance='no') a(1)-a(0)
  do i = 1, y-1
    write(*,'(x,i0)',advance='no') a(i+1)-a(i)
  end do
  write(*,*)
end program Sugoroku

Measured memory_bytes: 2212